


clear all
set more off
pause off
set matsize 11000



capture confirm file "Main_Results/Fig4/"
if _rc mkdir "Main_Results/Fig4/"

capture confirm file "Main_Results/Fig4/gphs/"
if _rc mkdir "Main_Results/Fig4/gphs/"




local region  Euro  

local response R_real




use "Data/Big_Pandemics.dta", clear
gen pandemic=1



drop year
ren endyear year
sort year
merge 1:1 year using "Data/RealRateSince1310.dta"
sort year
replace pandemic=0 if pandemic==.
gen pandemic100k=cond(dead>=100000,pandemic,0)


// drop big two and smallest six dummies
gen pandemic_noBig = pandemic100k
gen pandemic_noSmall= pandemic100k
replace pandemic_noBig = 0 if year==1352 | year==1920

// drop four smallest: first cholera, second cholera, Marseille, London and Sevilla
replace pandemic_noSmall = 0 if year==1652 | year==1666 | year==1722 | year==1826 | year==1851  | year==2009



cap drop Horizon
gen Horizon = _n-1 if _n<=41
tsset year


cap drop b0* se0* 
cap drop h0* l0* h1* l1*
tsset year
foreach x of local region {
	foreach y of local response {

	gen b0_`x'`y'=0 if Horizon==0
	gen se0_`x'`y'=0 if Horizon==0	
	gen h0_`x'`y'=0 if Horizon==0
	gen l0_`x'`y'=0 if Horizon==0
	gen h1_`x'`y'=0 if Horizon==0
	gen l1_`x'`y'=0 if Horizon==0
	
cap drop raw
gen raw = .
replace raw = `x'`y'


/********** State space specification starts here *************/

constraint 1	[rstar]L.rstar		= 1		/* Random Walk */	
constraint 2	[raw]rstar			= 1		/* Definitional */


cap drop zero								/* generate zero line for figures */
gen zero = 0

tsset year
cap drop r_star

/************	 State space equations	 ******************/

sspace		(rstar L.rstar, state noconstant)		/// /* r-star */
			(raw rstar, noconstant ),				/// /* raw */
			covstate(identity) constraints(1/2) difficult /*method(kdiffuse)*/ iterate(25)
			
			qui{
			/* Generate smoothed estimates of r-star */
			predict r_star if e(sample), states smethod(smooth) equation(rstar)
			
			
		
		
		
			cap drop `x'`y'`trend
			gen	`x'`y'trend = r_star
			}

			
			

								}
								
						}



// no Big Two pandemics				
						
foreach y of local response {
	foreach x of local region {
	eststo clear
		forv h=0/40 {
		
						cap drop  d`y'`h'
						gen d`y'`h' = f`h'.`x'`y'trend- l.`x'`y'trend
						reg d`y'`h'  l(0/10).pandemic_noBig l(2/11).`x'`y'trend  
												
						eststo
						replace b0_`x'`y'  = _b[pandemic_noBig]                      if Horizon==`h'
						replace se0_`x'`y' = _se[pandemic_noBig]                     if Horizon==`h'
						replace h0_`x'`y'  = _b[pandemic_noBig]+2*_se[pandemic_noBig]  if Horizon==`h'
						replace l0_`x'`y'  = _b[pandemic_noBig]-2*_se[pandemic_noBig]  if Horizon==`h'
						replace h1_`x'`y'  = _b[pandemic_noBig]+1*_se[pandemic_noBig]  if Horizon==`h'
						replace l1_`x'`y'  = _b[pandemic_noBig]-1*_se[pandemic_noBig]  if Horizon==`h'
						
						
					
						}
						
		esttab , keep(pandemic*)
		cap drop Zero
		gen Zero=0
		twoway ///
			(  rarea h0_`x'`y' l0_`x'`y' Horizon if Horizon~=. , fc(blue)  fi(10) lcolor(white)  lpattern(none)  )  ///
			(  rarea h1_`x'`y' l1_`x'`y' Horizon if Horizon~=. , fc(blue)  fi(20) lcolor(white) lw(none)  lpattern(none)  )  ///
			(  line  Zero    Horizon if Horizon~=. , lc(gray)  lw(med) lp(dash) )  ///		
			(  line  b0_`x'`y'     Horizon if Horizon~=. , lc(blue)  lw(thick) lp(solid)  ///
					xlab(0(10)40, labsize(large)) ylab(, angle(0) labsize(large)  nogrid) xtitle("Years since pandemic event ended",size(large)) ///
					ytitle("Percent",size(large)) ///
					legend( off ) )  ///
			, title()    ///
			xsize(5) ysize(4)  scheme(s1color)  ///
			name(`x'`y'_Big,replace) /// nodraw
	
		graph save   Main_Results/Fig4/gphs/Fig4c.gph, replace
		graph export Main_Results/Fig4/Fig4c.pdf, replace

}
}



// no smallest Six pandemics

cap drop b0* se0* 
cap drop h0* l0* h1* l1*
tsset year
foreach x of local region {
	foreach y of local response {
	gen b0_`x'`y'=0 if Horizon==0
	gen se0_`x'`y'=0 if Horizon==0	
	gen h0_`x'`y'=0 if Horizon==0
	gen l0_`x'`y'=0 if Horizon==0
	gen h1_`x'`y'=0 if Horizon==0
	gen l1_`x'`y'=0 if Horizon==0
	}
	}
	
foreach y of local response {
	foreach x of local region {
	eststo clear
		forv h=1/40 {
		
						cap drop  d`y'`h'
						gen d`y'`h' = f`h'.`x'`y'trend- l.`x'`y'trend
						reg d`y'`h'  l(0/10).pandemic_noSmall  l(2/11).`x'`y'trend  
												
						eststo
						replace b0_`x'`y'  = _b[pandemic_noSmall]                      if Horizon==`h'
						replace se0_`x'`y' = _se[pandemic_noSmall]                     if Horizon==`h'
						replace h0_`x'`y'  = _b[pandemic_noSmall]+2*_se[pandemic_noSmall]  if Horizon==`h'
						replace l0_`x'`y'  = _b[pandemic_noSmall]-2*_se[pandemic_noSmall]  if Horizon==`h'
						replace h1_`x'`y'  = _b[pandemic_noSmall]+1*_se[pandemic_noSmall]  if Horizon==`h'
						replace l1_`x'`y'  = _b[pandemic_noSmall]-1*_se[pandemic_noSmall]  if Horizon==`h'
						}
						
		esttab , keep(pandemic*)
		cap drop Zero
		gen Zero=0
		twoway ///
			(  rarea h0_`x'`y' l0_`x'`y' Horizon if Horizon~=. , fc(blue)  fi(10) lcolor(white)  lpattern(none)  )  ///
			(  rarea h1_`x'`y' l1_`x'`y' Horizon if Horizon~=. , fc(blue)  fi(20) lcolor(white) lw(none)  lpattern(none)  )  ///
			(  line  Zero    Horizon if Horizon~=. , lc(gray)  lw(med) lp(dash)                     )  ///		
			(  line  b0_`x'`y'     Horizon if Horizon~=. , lc(blue)  lw(thick) lp(solid)  ///
					xlab(0(10)40, labsize(large)) ylab(, angle(0) labsize(large)  nogrid) xtitle("Years since pandemic event ended", size(large)) ///
					ytitle("Percent",size(large)) ///
					legend( off ) )  ///
			, title()    ///
			xsize(5) ysize(4)  scheme(s1color) ///
			name(`x'`y'_Small,replace) /// nodraw
	
		graph save   Main_Results/Fig4/gphs/Fig4b.gph, replace
		graph export Main_Results/Fig4/Fig4b.pdf, replace

}
}



